

Arduino: Playground

The playground is a publicly-editable wiki about Arduino.

[Manuals and Curriculum](#)

[Board Setup and Configuration](#)

[Development Tools](#)

[Interfacing With Hardware](#)

- [Output](#)
- [Input](#)
- [Storage](#)
- [Communication](#)

[Interfacing with Software](#)

[Code Library and Tutorials](#)

[Suggestions & Bugs](#)

[Electronics Technique](#)

[Sources for Electronic Parts](#)

[Related Hardware and Initiatives](#)

[Arduino People/Groups & Sites](#)

[Exhibition](#)

[Languages](#)


PARTICIPATE

- [create an account](#)
- [suggestions](#)
- [formatting suggestions](#)
- [all recent changes](#)
- [PmWiki](#)
- [WikiSandBox training](#)
- [Basic Editing](#)
- [Cookbook \(addons\)](#)
- [Documentation Index](#)

[edit SideBar](#)

[View](#) [Edit](#) [History](#) [Print](#) [Login](#) [Logout](#)

Figure 1. H48C Connections



Keywords

Arduino Parallax H48C 3-axis acelerometer

description





FIGURE 5-1: Communication with the MCP3204 or MCP3208.

TABLE 5-1: CONFIGURATION BITS FOR THE MCP3204

Control Bit Selections				Input Configuration	Channel Selection
Single/Diff	D2*	D1	D0		
1	X	0	0	single-ended	CH0
1	X	0	1	single-ended	CH1
1	X	1	0	single-ended	CH2
1	X	1	1	single-ended	CH3
0	X	0	0	differential	CH0 = IN+ CH1 = IN-
0	X	0	1	differential	CH0 = IN- CH1 = IN+
0	X	1	0	differential	CH2 = IN+ CH3 = IN-
0	X	1	1	differential	CH2 = IN- CH3 = IN+

* D2 is a "don't care" for MCP3204

This sensor is controlled by a serial protocol as shown in figure 5.1

1. a start bit is send (function **StartBit()**)
2. command send to get the right data (function **ShiftOutNibble(Command)**) see table 5.1
3. The chip needs one CLK cycle to sample data, then the OUTPUT of DIO becomes INPUT (function **SampleIt()**)
4. The 12bit result is shifted in in three 4bit blocks (function **ShiftInNibble()**) and connected to one integer (range -2048 to 2048)
5. The sampling ends by deseleting the chip (function **EndBit()**)

The function **GetValue(byte Command)** includes everything in the right order:

```
ax = GetValue(B1000);
ay = GetValue(B1001);
az = GetValue(B1010);
```

CODE

```
/*
////////////////////////////////////
Htachi H48C3 Axis Accelerometer
parallax (#28026)

AUTHOR: kiilo kiilo@kiilo.org
License: http://creativecommons.org/licenses/by-nc-sa/2.5/ch/

http://parallax.com/Store/Microcontrollers/BASICStampModules/tabid/134/txtSearch/28026/List/1/ProductID/97/Default.aspx?SortField=ProductN
http://sage.medienkunst.ch/tiki-index.php?page=HowTo_Arduino_Parallax_H48C_Accelerometer
http://arduino.cc

////////////////////////////////////
*/
/// VARS
int CS_pin = 9;
int CLK_pin = 10;
int DIO_pin = 11;

int ax = 0;
int ay = 0;
int az = 0;

/// FUNCTIONS
void StartBit() {
  pinMode(DIO_pin, OUTPUT);
  digitalWrite(CS_pin, LOW);
  digitalWrite(CLK_pin, LOW);
  delayMicroseconds(1);
  digitalWrite(DIO_pin, HIGH);
  digitalWrite(CLK_pin, HIGH);
  delayMicroseconds(1);
}

void ShiftOutNibble(byte DataOutNibble) {
  for(int i = 3; i >= 0; i--) { // i = 3 ... 2 ... 1 ... 0
    digitalWrite(CLK_pin, LOW);
    // set DIO first
    if ((DataOutNibble & (1 << i)) == (1 << i)) { // DataOutNibble AND 1 x 2^i Equals 1 x 2^i ?
      digitalWrite(DIO_pin, HIGH);
    }
    else {
      digitalWrite(DIO_pin, LOW);
    }
    // with CLK rising edge the chip reads the DIO from arduino in
    digitalWrite(CLK_pin, HIGH);
    // data rate is f_clk 2.0 Mhz --> 0,5 micro seeconds
    delayMicroseconds(1); // :- ) just nothing
  }
}

void SampleIt() {
  digitalWrite(CLK_pin, LOW);
  delayMicroseconds(1);
  digitalWrite(CLK_pin, HIGH);
  delayMicroseconds(1);

  pinMode(DIO_pin, INPUT);
  digitalWrite(CLK_pin, LOW);
  delayMicroseconds(1);
  digitalWrite(CLK_pin, HIGH);
  if (digitalRead(DIO_pin) == LOW) {
    // Blink LED because ok
  }
}

byte ShiftInNibble() {
  byte resultNibble;
  resultNibble = 0;

  for(int i = 3; i >= 0; i--) { // from bit 3 to 0
    // The chip Shift out results on falling CLK
    digitalWrite(CLK_pin, LOW);
```

```

        delayMicroseconds(1); // :-) just nothing
        if( digitalRead(DIO_pin) == HIGH) { // BIT set or not?
            resultNibble += 1 << i; // Store 1 x 2^i in our ResultNibble
        }
        else {
            resultNibble += 0 << i; // YES this is always 0, just for symmetry :-)
        }
        digitalWrite(CLK_pin, HIGH);
        //delayMicroseconds(1); // :-) just nothing
    }
    return resultNibble;
}

void EndBit() {
    digitalWrite(CS_pin, HIGH);
    digitalWrite(CLK_pin, HIGH);
}

int GetValue(byte Command) { // x = B1000, y = B1001, z = B1010
    int Result = 0;
    StartBit();
    ShiftOutNibble(Command);
    SampleIt();
    Result = 2048 - ((ShiftInNibble() << 8) + (ShiftInNibble() << 4) + ShiftInNibble());
    EndBit();

    return Result;
}

//// SETUP
void setup() {
    Serial.begin(115200);
    pinMode(CS_pin, OUTPUT);
    pinMode(CLK_pin, OUTPUT);
    pinMode(DIO_pin, OUTPUT);
    // initialize device & reset
    digitalWrite(CS_pin,LOW);
    digitalWrite(CLK_pin,LOW);
    delayMicroseconds(1);
    digitalWrite(CS_pin, HIGH);
    digitalWrite(CLK_pin,HIGH);
}

//// LOOP
void loop() {

    aX = GetValue(B1000);
    aY = GetValue(B1001);
    aZ = GetValue(B1010);

    Serial.print(aX);
    Serial.print(" ");
    Serial.print(aY);
    Serial.print(" ");
    Serial.print(aZ);
    Serial.println("");
    delay(100); // loop every 10 times per sec.
}

```

Link

- [Parallax Product info H48C](#)
- [PDF Datasheet](#) see Chapt.5. serial communication protocoll P.15/16

Author

- [kiilo AT kiilo DOT org](#)